EXAMPLE 1.3-1: Magnetic Ablation

Thermal ablation is a technique for treating cancerous tissue that occurs by heating it to a lethal temperature. A number of techniques have been suggested to apply heat locally to the cancerous tissue and therefore spare surrounding healthy tissue. One interesting technique utilizes ferromagnetic thermoseeds, as discussed by Tompkins (1992). Small metallic spheres (thermoseeds) are embedded at precise locations within the cancer tumor and then the region is exposed to an oscillating magnetic field. The magnetic field does not generate thermal energy in the tissue; however, the spheres experience a volumetric generation of thermal energy which thereby increases their temperature and results in the conduction of heat to the surrounding tissue. Precise placement of the thermoseed can be used to control the application of thermal energy. The concept is shown in Figure 1.



Figure 1: A thermoseed used for ablation of a tumor.

It is necessary to determine the temperature field associated with a single thermoseed placed in an infinite medium of tissue. The thermoseed has a radius, r_{ts} = 1.0 mm and conductivity, k_{ts} = 10 W/m-K. A total of $\dot{g}_{ts}$ = 1.0 W of generation is uniformly distributed throughout the sphere. The temperature far from the thermoseed is the body temperature, T_b = 37°C. The tissue has thermal conductivity k_t = 0.5 W/m-K and is assumed to be in perfect thermal contact with the thermoseed. The effects of metabolic heat generation (i.e., volumetric generation in the tissue) and blood perfusion (i.e., the heat removed by blood flow in the tissue) are not considered in this problem.

a.) Prepare a plot showing the temperature in the thermoseed and in the tissue (i.e., the temperature from $r = 0$ to $r >> r_{ts}$).

This problem is 1-D because the temperature varies only in the radial direction, r; there are no circumferential non-uniformities in the problem that would lead to temperature gradients in any dimension except r. However, the problem includes two, separate computational domains that share a common boundary (i.e., the thermoseed and the tissue). Thus there will be two different governing equations that must be solved and additional boundary conditions that must be considered. The solution will require somewhat more algebra; fortunately, this additional algebra is not a concern if the problem is solved using EES.

It is always good to start your problem with an input section in which all of the given information is entered and, if necessary, converted to SI units.

```
"EXAMPLE 1.3-1: Magnetic Ablation"
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in

"Inputs"
r_ts=1 [mm]*convert(mm,m)			"radius of the thermoseed"
k_ts=10 [W/m-K]				"thermal conductivity of thermoseed"
g_dot_ts=1.0 [W]				"total generation of thermal energy in thermoseed"
T_b=converttemp(C,K,37 [C])			"body temperature"
k_t=0.5 [W/m-K]				"tissue thermal conductivity"
```

Notice a few things about the EES code above. First, comments are provided to define the nomenclature and make the code understandable; this type of annotation is important for clarity and organization. Also, units are not ignored but rather explicitly specified and dealt with as the problem is set up, rather than as an afterthought at the end. The unit system EES will use can be specified in the Properties Dialog (select Preferences from the Options menu) from the Unit System tab or using the $UnitSystem directive, as shown in the EES code above. The units of numerical constants can be set directly in square brackets following the value. For example, the statement

```
r_ts=1 [mm]*convert(mm,m)	"radius of the thermoseed"
```

tells EES that the constant 1 has units of mm and these should be converted to units of m; therefore the variable r_ts should have units of m. The units of r_ts are not automatically set, as the equation involving r_ts is not a simple assignment. If you check units at this point (select Check Units from the Calculate menu) then EES will indicate that there is a unit conversion error. This unit error occurs because the variable r_ts has not been assigned any units but the equations are consistent with r_ts having units of m. It is possible to have EES set units automatically (this is an option in the Options tab in the Properties Dialog); however, this is not recommended because the engineer doing the problem should know and set the units for each variable.

The Formatted Equations window (select Formatted Equations from the Windows menu) shows the equations and their units more clearly (Figure 2).



Figure 2: Formatted Equations window.

It is good practice to set the units of all variables. One method of accomplishing this is to right-click on a variable in the Solutions window which brings up the Format Selected Variables dialog; the units can be typed directly into the Units input box. Note that right-clicking in the Units input box and selecting the Unit List menu item provides a partial list of the SI units that EES recognizes. All of the units that are recognized by EES can be examined by selecting Unit Conversion Info from the Options menu.

Once the units of r_ts are set to m, then a unit check (select Check Units from the Calculate menu) should reveal no errors.

It is necessary to work with a different governing equation in each of the two computational domains. An appropriate differential control volume for the spherical geometry with uniform generation was presented in Section 1.3.3 and led to the general solution listed in Table 1- 3. The general solution that is valid within the thermoseed (i.e., from $0 < r < r_{ts}$) is:

$$T_{ts} = -\frac{\dot{g}'''_{ts}}{6\,k_{ts}} r^2 + \frac{C_1}{r} + C_2 \qquad (1)$$

where C_1 and C_2 are undetermined constants of integration and $\dot{g}'''_{ts}$ is the volumetric rate of generation of thermal energy in the thermoseed, which can be expressed as the ratio of the total rate of thermal energy generation to the volume of the thermoseed:

$$\dot{g}'''_{ts} = \frac{3\,\dot{g}_{ts}}{4\,\pi\,r_{ts}^3}$$

```
g```_dot_ts=3*g_dot_ts/(4*pi*r_ts^3)       "volumetric rate of generation in the thermoseed"
```

The general solution that is valid within the tissue (i.e., for $r > r_{ts}$) is:

$$T_t = \frac{C_3}{r} + C_4 \qquad (2)$$

because the volumetric rate of thermal energy generation in the tissue is zero; note that C_3 and C_4 are different undetermined constants of integration than C_1 and C_2 in Eq. (1).

The next step is to define the boundary conditions. There are four undetermined constants and therefore there must be four boundary conditions. At the center of the thermoseed (r = 0) the temperature must remain finite. Substituting r = 0 into Eq. (1) leads to:

$$T_{ts,r=0} = -\frac{0^2}{6\,k_{ts}}\dot{g}'''_{ts} + \frac{C_1}{k_{ts}\,0} + C_2$$

which indicates that C_1 must be zero:

$$C_1 = 0 \tag{3}$$

Alternatively, specifying that the temperature gradient at the center of the sphere is equal to zero (consistent with zero heat transfer rate at the sphere center) leads to the same conclusion, $C_1 = 0$. As the radius approaches infinity, the tissue temperature must approach the body temperature. Substituting $r \rightarrow \infty$ into Eq. (2) leads to:

$$T_b = -\frac{C_3}{\infty} + C_4$$

which indicates that:

$$C_4 = T_b \tag{4}$$

The remaining boundary conditions are defined at the interface between the thermoseed and the tissue. It is assumed that the sphere and tissue are in perfect thermal contact (i.e., there is no contact resistance at this interface) so that the temperature must be continuous at the interface; therefore:

$$T_{ts,r=r_{ts}} = T_{t,r=r_{ts}}$$

or, substituting $r = r_{ts}$ into Eqs. (1) and (2):

$$-\frac{r_{ts}^2}{6\,k_{ts}}\dot{g}'''_{ts} + \frac{C_1}{r_{ts}} + C_2 = \frac{C_3}{r_{ts}} + C_4 \tag{5}$$

An energy balance on the interface (see Figure 1) requires that the steady-state heat transfer rate at the outer edge of the thermoseed ($\dot{q}_{r=r_{ts}^-}$ in Figure 1) must equal the heat transfer rate at the inner edge of the tissue ($\dot{q}_{r=r_{ts}^+}$ in Figure 1).

$$\dot{q}_{r=r_{ts}^-} = \dot{q}_{r=r_{ts}^+} \tag{6}$$

According to Fourier's law, Eq. (6) can be written as:

$$-4\,\pi\, r_{ts}^2\, k_{ts} \left.\frac{dT_{ts}}{dr}\right|_{r=r_{ts}} = -4\,\pi\, r_{ts}^2\, k_t \left.\frac{dT_t}{dr}\right|_{r=r_{ts}}$$

or

$$k_{ts} \left.\frac{dT_{ts}}{dr}\right|_{r=r_{ts}} = k_t \left.\frac{dT_t}{dr}\right|_{r=r_{ts}} \tag{7}$$

Substituting Eqs. (1) and (2) into Eq. (7) leads to:

$$-k_{ts}\left(-\frac{r_{ts}}{3\,k_{ts}}\dot{g}_{ts}''' - \frac{C_1}{r_{ts}^2}\right) = -k_t\left(-\frac{C_3}{r_{ts}^2}\right) \tag{8}$$

Entering Eqs. (3), (4), (5), and (8) into EES will lead to the solution of the four constants without algebra and the associated opportunities for error.

```
"Determine constants of integration"
C_1=0                                                      "temperature at center must be finite"
C_4=T_b                                                    "temperature far from the thermoseed"
-r_ts^2*g```_dot_ts/(6*k_ts)+C_1/r_ts+C_2=C_3/r_ts+C_4     "continuity of temperature at the interface"
-k_ts*(-r_ts*g```_dot_ts/(3*k_ts)-C_1/r_ts^2)=-k_t*(-C_3/r_ts^2) "equal heat flux at the interface"
```

Finally, we can generate a plot using the solution. Displaying the radius in millimeters in the plot will make a much more reasonable scale than in meters. A new variable, r_mm, is defined for this purpose and the solution is converted from K to C.

```
"Prepare a plot"
r=r_mm*convert(mm,m)                          "radius"
T_ts=-r^2*g```_dot_ts/(6*k_ts)+C_1/r+C_2      "thermoseed temperature"
T_ts_C=converttemp(K,C,T_ts)                  "in C"
T_t=C_3/r+C_4                                 "tissue temperature"
T_t_C=converttemp(K,C,T_t)                    "in C"
```

The units of the variables C_1, C_2, etc. should be set in the Variable Information window and the set of equations subsequently checked for unit consistency. The relationship between temperature and radial position will be determined using two Parametric Tables. The first table will include variables r_mm and T_ts_C and the second will include the variables r_mm and T_t_C. In the first table, r_mm is varied from 0 to 1.0 mm (i.e., within the thermoseed) and in the second it is varied from 1.0 mm to 10.0 mm (i.e., within the tissue). A plot in which the data contained in the two tables is overlaid leads to the temperature distribution, shown in Figure 3.



Figure 3: Temperature distribution through sphere and tissue.

Figure 3 agrees with physical intuition; the temperature decays towards the body temperature with increasing distance from the sphere. The rate of energy being transferred from the sphere through the tissue is constant but the area for conduction is growing as r^2 and therefore the gradient in the temperature is dropping. The conduction heat transfer rates at the outer edge of the sphere (i.e., $r = r_{sp}^-$) and at the inner edge of the tissue (i.e., $r = r_{sp}^+$) are identical; the discontinuity in slope is related to the fact that the thermoseed is more conductive than the tissue.

b.) Determine the maximum temperature in the tissue and the extent of the lesion as a function of the rate of thermal energy generation in the thermoseed. The extent of the lesion (r_{lesion}) is defined as the radial location where the tissue temperature reaches the lethal temperature for tissue, approximately T_{lethal} = 50°C according to Izzo, (2003).

The maximum tissue temperature ($T_{t,max}$) is the temperature at the interface between the thermoseed and the tissue and is obtained by substituting $r = r_{ts}$ into Eq. (2):

$$T_{t,max} = \frac{C_3}{r_{ts}} + C_4$$

```
T_t_max=C_3/r_ts+C_4                          "maximum tissue temperature"
T_t_max_C=converttemp(K,C,T_t_max)            "in C"
```

The extent of the lesion can be obtained by determining the radial location where $T_t = T_{lethal}$:

$$T_{lethal} = \frac{C_3}{r_{lesion}} + C_4$$

```
T_lethal=converttemp(C,K, 50 [C])             "lethal temperature for cell death"
T_lethal=C_3/r_lesion+C_4                     "determine the extent of the lesion"
r_lesion_mm=r_lesion*convert(m,mm)            "in mm"
```

In order to investigate the maximum tissue temperature and the extent of the lesion as a function of the thermal energy generation rate, it is necessary to prepare a Parametric table that includes the variables T_t_max_C, r_lesion, and g_dot_sp and vary g_dot_sp within the table. Figure 4 illustrates the maximum temperature and the lesion extent as a function of the rate of thermal energy generation in the thermoseed. Note that $T_{t,max}$ and r_{lesion} have very different magnitudes and therefore it is necessary to plot the variable r_lesion on a secondary y-axis.



Figure 4: Maximum tissue temperature and the extent of lesion as a function of the rate of thermal energy generation in the thermoseed.